
1 Business Scenario 
With performance-critical extractions, several InfoPackages must be started in parallel. Parallel extraction 
must be enabled using disjointed selection criteria. An ABAP program is used to generate InfoPackages 
for a specific DataSource. The selection parameters are filled at runtime. 
 
The selection parameters for the InfoPackages are filled from field CARRID of table SFLIGHT and have 
the following content: 
 

 
 The result is two InfoPackages with the selection criteria “LH” and “SQ”. 

Result 

InfoPackage 1: 

 
 
InfoPackage 2: 

 

2 Introduction 
To manually create InfoPackages requires a high user workload. Incorrect selection criteria can cause 
data inconsistencies, whether it be the result of a material number being omitted, or different selection 
criteria overlapping.  
 
By using BAPIs to delete, create and change InfoPackages, this process can be automated in an ABAP 
program. 

3 The Step By Step Solution 
The staging process must have already been created.  

1. Requirements 
 

Define the InfoSource & selection 
criteria for which the InfoPackages are 
to be created. 
 
A generic DataSource is used in this 
example. 
 
DataSource: 
ZHOWTOINFOPACKAGEBAPI 
 
InfoSource: 
ZHOWTOINFOPACKAGEBAPI 
 

 

 
 
 

 
 

1. 

2. BAPI Used 
 

1. You can find all BAPIs that exist in the 
system via transaction BAPI. 
 
In this example, the following BAPIs will 
be used: 
 
BAPI_IPAK_GETLIST 
BAPI_IPAK_CREATE 
BAPI_IPAK_DELETE 
 
These are also function modules at the 
same time. 

3. Program to Delete and Create InfoPackages 
 

2. Via transaction SE38, create an 
executable program. 
 
In this example, the program is called: 
 
ZAUTO_CREATE_AND_DELETE_IP 

 

 
 

3. Specify the properties of the program 
and save.  

In the following program section: 
 
******************************************* 
create InfoPackages with SELECTION-CRITERIA 

******************************************* 

the BAPI interface parameters are adjusted. The sample code from the example can be used to get 
the suitable parameters.  

Code to generate InfoPackages: 
*&---------------------------------------------------------------------* 
*& Report  ZAUTO_CREATE_AND_DELETE_IP                                  * 
*&                                                                     * 
*&---------------------------------------------------------------------* 
*&                                                                     * 
*&                                                                     * 
*&---------------------------------------------------------------------* 
 
REPORT  ZAUTO_CREATE_AND_DELETE_IP              . 
************************************************************************ 
* DATA Declaration 
************************************************************************ 
* GET InfoPackage 
DATA: SO_S_SELINFOSOURCE     type BAPI6102SL, 
      SO_S_SELSOURCESYSTEM   type BAPI6101SL, 
      SO_S_SELDATASOURCE     type BAPI6109DSSL, 
       I_S_INFOPACKAGE_LIST  type BAPI6109L, 
       I_S_RETURN            type BAPIRET2. 
 
DATA: SO_T_SELINFOSOURCE     type BAPI6102SL   occurs 0, 
      SO_T_SELSOURCESYSTEM   type BAPI6101SL   occurs 0, 
      SO_T_SELDATASOURCE     type BAPI6109DSSL occurs 0, 
       I_T_INFOPACKAGE_LIST  type BAPI6109L    occurs 0, 
       I_T_RETURN            type BAPIRET2     occurs 0. 
 
* CREATE InfoPackage 
DATA: I_S_DETAILS            type BAPI6109, 
      I_S_SCHEDULING_PARAMS  type BAPI6109BTCH, 
      I_S_DESTINATIONS       type BAPI6109DEST, 
      P_INFOPACKAGE          type BAPI6109-INFOPACKAGE, 
      I_S_SELECTIONS         type BAPI6109SEL. 
 
DATA: I_T_SELECTIONS         type BAPI6109SEL occurs 0. 
 
 
 
************************************************************************ 
* get all avaiable InfoPackages for a special InfoSource 
* Fill SELECT-OPTION's to get the InfoPackage List 
************************************************************************ 
* InfoSource 
move 'I'                     TO SO_S_SELINFOSOURCE-SIGN. 
move 'EQ'                    TO SO_S_SELINFOSOURCE-OPTION. 
move 'ZHOWTOINFOPACKAGEBAPI' TO SO_S_SELINFOSOURCE-INFOSOURCELOW. 
move ''                      TO SO_S_SELINFOSOURCE-INFOSOURCEHIGH. 
APPEND SO_S_SELINFOSOURCE    TO SO_T_SELINFOSOURCE. 
 
* SourceSystem 

move 'I'                     TO SO_S_SELSOURCESYSTEM-SIGN. 
move 'EQ'                    TO SO_S_SELSOURCESYSTEM-OPTION. 
move 'QB8CLNT003'            TO SO_S_SELSOURCESYSTEM-SOURCESYSTEMLOW. 
move ''                      TO SO_S_SELSOURCESYSTEM-SOURCESYSTEMHIGH. 
 
APPEND SO_S_SELSOURCESYSTEM  TO SO_T_SELSOURCESYSTEM. 
* DataSource 
move 'I'                     TO SO_S_SELDATASOURCE-SIGN. 
move 'EQ'                    TO SO_S_SELDATASOURCE-OPTION. 
move 'ZHOWTOINFOPACKAGEBAPI' TO SO_S_SELDATASOURCE-DATASOURCELOW. 
move ''                      TO SO_S_SELDATASOURCE-DATASOURCEHIGH. 
APPEND SO_S_SELDATASOURCE    TO SO_T_SELDATASOURCE. 
 
* Call BAPI BAPI_IPAK_GETLIST and get the List 
CALL FUNCTION 'BAPI_IPAK_GETLIST' 
* EXPORTING 
*   JOB_STATUS             = ' ' 
  TABLES 
*   SELTEXTLONG            = 
    SELINFOSOURCE          = SO_T_SELINFOSOURCE 
    SELSOURCESYSTEM        = SO_T_SELSOURCESYSTEM 
    SELDATASOURCE          = SO_T_SELDATASOURCE 
    INFOPACKAGE_LIST       =  I_T_INFOPACKAGE_LIST 
    RETURN                 =  I_T_RETURN. 
 
if sy-subrc <> 0. 
  "     error handling 
endif. 
 
************************************************************************ 
* delete all existing InfoPackages 
************************************************************************ 
if I_T_INFOPACKAGE_LIST is not initial and I_T_RETURN is initial. 
 
  loop at I_T_INFOPACKAGE_LIST into I_S_INFOPACKAGE_LIST. 
 
*   call BAPI to DELETE InfoPackage 
    CALL FUNCTION 'BAPI_IPAK_DELETE' 
      EXPORTING 
        INFOPACKAGE = I_S_INFOPACKAGE_LIST-INFOPACKAGE 
      TABLES 
        RETURN      = I_T_RETURN. 
 
    if sy-subrc <> 0 or I_T_RETURN is not initial. 
      "     error handling 
    endif. 
 
  endloop. 
endif. 
 
 
************************************************************************ 
* create InfoPackages with SELECTION-CRITERIA 
************************************************************************ 
 
* get SELECTION-CRITERIA from table SFLIGHT and field CARRID 
DATA: BEGIN OF I_S_SFLIGHT, 
        CARRID TYPE SFLIGHT-CARRID, 
      END OF I_S_SFLIGHT. 
 
DATA: I_T_SFLIGHT LIKE I_S_SFLIGHT occurs 0. 
 
clear   I_S_SFLIGHT. 
refresh I_T_SFLIGHT. 

select distinct carrid from sflight into table I_T_SFLIGHT. 
 
* loop all CARRID enties from table SFLIGHT 
loop at i_t_sflight into i_s_sflight. 
  clear   I_S_SELECTIONS. 
  refresh I_T_SELECTIONS. 
 
  " structure I_S_DETAILS 
  move 'ZHOWTOINFOPACKAGEBAPI' to I_S_DETAILS-INFOSOURCE. 
  move 'I'                     to I_S_DETAILS-INFOSOURCETYPE. 
  move 'QB8CLNT003'            to I_S_DETAILS-LOGSYS. 
  move 'ZHOWTOINFOPACKAGEBAPI' to I_S_DETAILS-DATASOURCE. 
  move 'D'                     to I_S_DETAILS-DATASOURCETYPE. 
  move '0'                     to I_S_DETAILS-HIERACT. 
  move ''                      to I_S_DETAILS-CHAVL_CHECK. 
  move 'X'                     to I_S_DETAILS-UPDATEMASTERDATA. 
  " generate short text of InfoPackage 
  data: tabix type string. 
  move sy-tabix to tabix. 
  concatenate sy-datum sy-uzeit 'InfoPackage : ' tabix into 
  I_S_DETAILS-TEXT separated by space. 
  move ''                      to I_S_DETAILS-LOAD_STATUS. 
  move 'F'                     to I_S_DETAILS-UPDMODE_READ. 
 
  " structure I_S_SCHEDULING_PARAM 
  move '' to I_S_SCHEDULING_PARAMS-STARTDTTYP. 
  move '' to I_S_SCHEDULING_PARAMS-SDLSTRTDT. 
  move '' to I_S_SCHEDULING_PARAMS-SDLSTRTTM. 
  move '' to I_S_SCHEDULING_PARAMS-EVENTID. 
  move '' to I_S_SCHEDULING_PARAMS-EVENTPARM. 
  move '' to I_S_SCHEDULING_PARAMS-PERIODIC. 
  move '' to I_S_SCHEDULING_PARAMS-CANCELPERIOD. 
  move '' to I_S_SCHEDULING_PARAMS-EVENT_ID_OK. 
  move '' to I_S_SCHEDULING_PARAMS-EVENT_PM_OK. 
  move '' to I_S_SCHEDULING_PARAMS-EVENT_ID_ER. 
  move '' to I_S_SCHEDULING_PARAMS-EVENT_PM_ER. 
  move '' to I_S_SCHEDULING_PARAMS-FUNCTION_CALL. 
  move '' to I_S_SCHEDULING_PARAMS-EXIT_CALL. 
  move '' to I_S_SCHEDULING_PARAMS-BATCH_WAIT. 
  move '' to I_S_SCHEDULING_PARAMS-JOBNAME. 
 
  " structure I_S_DESTINATIONS 
  move 'X' to I_S_DESTINATIONS-ONLYENTRANCE. 
  move ''  to I_S_DESTINATIONS-ONLYCUBETAB. 
  move ''  to I_S_DESTINATIONS-PARALLEL. 
  move ''  to I_S_DESTINATIONS-CUBEAFTERODS. 
 
  " table selections 
  move 'CARRID'           to I_S_SELECTIONS-FIELDNAME. 
  move 'CARRID'           to I_S_SELECTIONS-IOBJNM. 
  move 'I'                to I_S_SELECTIONS-SIGN. 
  move 'EQ'               to I_S_SELECTIONS-OPT. 
  move I_S_SFLIGHT-CARRID to I_S_SELECTIONS-LOW. 
  append I_S_SELECTIONS   to I_T_SELECTIONS. 
 
* create InfoPackages with BAPI 
  CALL FUNCTION 'BAPI_IPAK_CREATE' 
    EXPORTING 
      DETAILS                  = I_S_DETAILS 
      SCHEDULING_PARAMS        = I_S_SCHEDULING_PARAMS 
      DESTINATIONS             = I_S_DESTINATIONS 
*     FILE_PARAMS              = 
*     HIE_PARAMS               = 

    IMPORTING 
      INFOPACKAGE              = P_INFOPACKAGE 
    TABLES 
      SELECTIONS               = I_T_SELECTIONS 
*     INFOCUBES                = 
*     THIRD_PARTY_PARAMS       = 
      RETURN                   = I_T_RETURN. 
 
  if sy-subrc <> 0 or I_T_RETURN is not initial. 
*  error handling 
  endif. 
 
endloop. 

4 Appendix (Sample ABAP Coding)   
In order to fill interface BAPI_IPAK_CREATE correctly, it is advisable to read an existing InfoPackage with 
interface BAPI_IPAK_GETDETAIL and to analyze the returned parameters in the debugger. 
The following coding: 
 
 
 
*********************************************************************** 
* 
* BEGIN APPENDIX 
* get InfoPackage Details EXAMPLE 
*********************************************************************** 
 
 
P_INFOPACKAGE = 'XXXXXXXXXXXXXXXX'." technical Name of InfoPackage 
 
DATA: I_T_INFOCUBES          type BAPI6109IC occurs 0, 
      I_T_THIRD_PARTY_PARAMS type BAPI6109TCP occurs 0, 
      I_S_FILE_PARAMS        type BAPI6109FILE, 
      I_S_HIE_PARAMS         type BAPI6109HIE. 
DATA: I_S_DETAILS            type BAPI6109, 
      I_S_SCHEDULING_PARAMS  type BAPI6109BTCH, 
      I_S_DESTINATIONS       type BAPI6109DEST, 
      P_INFOPACKAGE          type BAPI6109-INFOPACKAGE, 
      I_S_SELECTIONS         type BAPI6109SEL. 
 
DATA: I_T_SELECTIONS         type BAPI6109SEL occurs 0. 
 
 
* fill BAPI Interface 
CALL FUNCTION 'BAPI_IPAK_GETDETAIL' 
  EXPORTING 
    INFOPACKAGE        = P_INFOPACKAGE 
  IMPORTING 
    DETAILS            = I_S_DETAILS 
    SCHEDULING_PARAMS  = I_S_SCHEDULING_PARAMS 
    DESTINATIONS       = I_S_DESTINATIONS 
    FILE_PARAMS        = I_S_FILE_PARAMS 
    HIE_PARAMS         = I_S_HIE_PARAMS 
  TABLES 
    SELECTIONS         = I_T_SELECTIONS 
    INFOCUBES          = I_T_INFOCUBES 
    THIRD_PARTY_PARAMS = I_T_THIRD_PARTY_PARAMS 
    RETURN             = I_T_RETURN. 

break-point. 
 
*********************************************************************** 
* END APPENDIX 

*********************************************************************** 

